

# Unit Checked Arithmetic via Unitful.jl

Chris Rackauckas

March 7, 2019

Units and dimensional analysis are standard tools across the sciences for checking the correctness of your equation. However, most ODE solvers only allow for the equation to be in dimensionless form, leaving it up to the user to both convert the equation to a dimensionless form, punch in the equations, and hopefully not make an error along the way.

DifferentialEquations.jl allows for one to use Unitful.jl to have unit-checked arithmetic natively in the solvers. Given the dispatch implementation of the Unitful, this has little overhead.

## 0.1 Using Unitful

To use Unitful, you need to have the package installed. Then you can add units to your variables. For example:

```
using Unitful  
t = 1.0u"s"
```

1.0 s

Notice that `t` is a variable with units in seconds. If we make another value with seconds, they can add

```
t2 = 1.02u"s"  
t+t2
```

2.02 s

and they can multiply:

```
t*t2
```

1.02 s<sup>2</sup>

You can even do rational roots:

```
sqrt(t)
```

```
1.0 s1/2
```

Many operations work. These operations will check to make sure units are correct, and will throw an error for incorrect operations:

```
t + sqrt(t)
```

```
Error: DimensionError: 1.0 s and 1.0 s1/2 are not dimensionally compatible.
```

## 0.2 Using Unitful with DifferentialEquations.jl

Just like with other number systems, you can choose the units for your numbers by simply specifying the units of the initial condition and the timestep. For example, to solve the linear ODE where the variable has units of Newton's and `t` is in Seconds, we would use:

```
using DifferentialEquations
f = (y,p,t) -> 0.5*y
u0 = 1.5u"N"
prob = ODEProblem(f,u0,(0.0u"s",1.0u"s"))
sol = solve(prob,Tsit5())
```

```
Error: DimensionError: N s-1 and 0.75 N are not dimensionally compatible.
```

Notice that we recieved a unit mismatch error. This is correctly so! Remember that for an ODE:

$$\frac{dy}{dt} = f(t, y)$$

we must have that `f` is a rate, i.e. `f` is a change in `y` per unit time. So we need to fix the units of `f` in our example to be N/s. Notice that we then do not receive an error if we do the following:

```
f = (y,p,t) -> 0.5*y/3.0u"s"
prob = ODEProblem(f,u0,(0.0u"s",1.0u"s"))
sol = solve(prob,Tsit5())
```

```
retcode: Success
Interpolation: specialized 4th order "free" interpolation
t: 3-element Array{Unitful.Quantity{Float64,T,Unitful.FreeUnits{(s,),T,nothing}},1}:
```