

Optimizing DiffEq Code

Chris Rackauckas

July 4, 2020

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.83 MiB
  allocs estimate: 101435
  -----
  minimum time:      3.460 ms (0.00% GC)
  median time:       3.492 ms (0.00% GC)
  mean time:         4.427 ms (16.36% GC)
  maximum time:      9.929 ms (46.35% GC)
  -----
  samples:           1129
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.47 MiB
  allocs estimate: 88655
  -----
  minimum time:      2.933 ms (0.00% GC)
  median time:       2.957 ms (0.00% GC)
  mean time:         3.904 ms (15.77% GC)
  maximum time:      9.495 ms (48.00% GC)
  -----
  samples:           1280
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to *****make our code non-allocating*****:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
  du[2] = u[1]*(28.0-u[3]) - u[2]
  du[3] = u[1]*u[2] - (8/3)*u[3]
end

lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes

the internal allocations as well. When we benchmark this function, we will see quite a difference.

```
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  1.38 MiB
  allocs estimate: 12997
  -----
  minimum time:     780.565 μs (0.00% GC)
  median time:      791.816 μs (0.00% GC)
  mean time:        904.446 μs (9.93% GC)
  maximum time:     5.962 ms (81.26% GC)
  -----
  samples:          5516
  evals/sample:     1
```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.27 KiB
  allocs estimate:  57
  -----
  minimum time:     342.286 μs (0.00% GC)
  median time:      345.266 μs (0.00% GC)
  mean time:        345.623 μs (0.00% GC)
  maximum time:     378.963 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1
```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```
tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.27 KiB
  allocs estimate:  57
  -----
  minimum time:     1.716 ms (0.00% GC)
  median time:      1.725 ms (0.00% GC)
  mean time:        1.726 ms (0.00% GC)
  maximum time:     1.901 ms (0.00% GC)
  -----
  samples:          2894
  evals/sample:     1
```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are](#)

a lot of sources online. But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. `structs` for example are stack-allocated "value-type"s. `Tuples` are a stack-allocated collection. The most useful data structure for `DiffEq` though is the `StaticArray` from the package `StaticArrays.jl`. These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```
using StaticArrays
A = @SVector [2.0,3.0,5.0]

3-element StaticArrays.SArray{Tuple{3},Float64,1,3} with indices SOneTo(3):
 2.0
 3.0
 5.0
```

Notice that the 3 after `SVector` gives the size of the `SVector`. It cannot be changed. Additionally, `SVectors` are immutable, so we have to create a new `SVector` to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. `SArrays` have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons. Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's `***optimize lorenz using static arrays***`. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end

lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
  memory estimate:  466.30 KiB
  allocs estimate:  2596
  -----
  minimum time:     338.074 μs (0.00% GC)
  median time:      344.546 μs (0.00% GC)
  mean time:        375.793 μs (5.46% GC)
  maximum time:     3.804 ms (87.13% GC)
  -----
  samples:          10000
  evals/sample:     1

@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  3.44 KiB
  allocs estimate:  32
  -----
  minimum time:     238.745 μs (0.00% GC)
  median time:      242.229 μs (0.00% GC)
  mean time:        242.369 μs (0.00% GC)
  maximum time:     285.171 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1

```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```

A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:     1.122 ms (0.00% GC)
  median time:      1.150 ms (0.00% GC)
  mean time:        1.305 ms (11.73% GC)

```

```

maximum time:      3.112 ms (60.47% GC)
-----
samples:           3814
evals/sample:      1

```

That expression $A + B + C$ creates 2 arrays. It first creates one for the output of $A + B$, then uses that result array to $+ C$ to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the $+$ operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```

test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  8
-----
 minimum time:     2.173 ms (0.00% GC)
 median time:     2.201 ms (0.00% GC)
 mean time:       2.341 ms (5.98% GC)
 maximum time:     3.984 ms (44.16% GC)
-----
 samples:          2134
 evals/sample:     1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
 minimum time:     1.121 ms (0.00% GC)
 median time:     1.143 ms (0.00% GC)
 mean time:       1.292 ms (11.48% GC)
 maximum time:     3.002 ms (60.64% GC)
-----
 samples:          3851
 evals/sample:     1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2

```

```

-----
minimum time:      1.118 ms (0.00% GC)
median time:       1.145 ms (0.00% GC)
mean time:         1.296 ms (11.45% GC)
maximum time:      3.027 ms (59.68% GC)
-----

samples:           3841
evals/sample:      1

```

is a version with only 1 array created (the output). Note that `.s` can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```

1000×1000 Array{Float64,2}:
 0.513261  1.31848    1.2224    0.255684  ...  0.380879  0.639533  0.649082
 1.40662   1.41229    0.758463  0.596732  ...  0.547179  0.740511  0.158411
 1.3567    0.984284    1.175     0.85477   ...  0.756969  0.534338  0.320014
 0.967212  1.33246    0.557833  0.739195  ...  1.49634   1.58451   1.12776
 0.911932  1.45992    0.891     0.76021   ...  0.804562  1.05281   1.1543
 1.41783   0.459628    0.87301   0.612895  ...  0.691527  1.12037   0.348726
 0.654021  0.880541    0.501332  1.38162   ...  1.18179   0.872631  0.912316
 1.32338   0.126603    0.821788  0.837101  ...  1.17346   1.38669   0.960177
 1.22324   1.26655     1.21557   0.850047  ...  1.23699   0.715315  0.899733
 0.593026  1.34217     0.660458  0.416883  ...  1.40208   1.11839   0.920313
 ⋮
 0.610308  1.20222     0.504438  1.05008   ...  1.08323   0.178886  1.46929
 1.06426   1.59739     0.570027  0.405927  ...  0.826415  1.14532   1.41054
 0.857372  0.868324    0.287592  1.01889   ...  0.538279  0.894807  0.710117
 0.926426  1.63151     0.379539  1.08774   ...  1.35114   0.16207   0.555124
 1.2718    1.3275     0.907245  1.09218   ...  0.856339  0.578431  0.569911
 0.733894  1.12087     0.513927  0.856334  ...  1.26028   0.820517  1.04967
 0.866924  0.943234    0.117487  1.43045   ...  0.754101  1.14463   0.827323
 1.57794   0.86349     1.3256    0.883278  ...  0.830822  0.375132  0.390161
 0.314827  0.0943557    0.76722   0.302452  ...  0.918325  0.674099  1.30896

```

Also, the `@.` macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:      1.125 ms (0.00% GC)
median time:       1.151 ms (0.00% GC)
mean time:         1.300 ms (11.44% GC)
maximum time:      3.054 ms (58.98% GC)
-----

samples:           3828
evals/sample:      1

```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via `.=`. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

[illegible]

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:

```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.040 ms (0.00% GC)
  median time:       1.046 ms (0.00% GC)
  mean time:         1.049 ms (0.00% GC)
  maximum time:      1.336 ms (0.00% GC)
  -----
  samples:           4742
  evals/sample:      1
```


If we use `@.` before the `=`, then it will turn it into `.=`:

```
test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.041 ms (0.00% GC)
  median time:       1.050 ms (0.00% GC)
  mean time:         1.052 ms (0.00% GC)
  maximum time:      1.339 ms (0.00% GC)
  -----
  samples:           4729
  evals/sample:      1
```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```
test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 32 bytes
  allocs estimate: 1
  -----
  minimum time:      2.282 ms (0.00% GC)
  median time:       2.291 ms (0.00% GC)
  mean time:         2.296 ms (0.00% GC)
  maximum time:      2.543 ms (0.00% GC)
  -----
  samples:           2174
  evals/sample:      1
```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```
BenchmarkTools.Trial:
  memory estimate: 7.63 MiB
  allocs estimate: 2
  -----
  minimum time:      6.404 ms (0.00% GC)
  median time:       6.482 ms (0.00% GC)
  mean time:         6.646 ms (2.12% GC)
  maximum time:      8.474 ms (21.50% GC)
  -----
  samples:           752
  evals/sample:      1
```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```
using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B
```

```

BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      5.975 ms (0.00% GC)
  median time:       6.011 ms (0.00% GC)
  mean time:         6.018 ms (0.00% GC)
  maximum time:      6.863 ms (0.00% GC)
  -----
  samples:           830
  evals/sample:      1

```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, **we can fuse together higher level linear algebra operations using BLAS**. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{a u^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + a u^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```

# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal}([1.0 for i in 1:N-1],[-2.0 for i in 1:N],[1.0 for i in 1:N-1]))
Ay = copy(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du .+ a.*u.*u./v .+ ubar .- α*u
    dr[:, :, 2] = Dv .+ a.*u.*u .- β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros(100,100,2)

```

```

r0[:, :, 1] .= u0 + 0.1 * rand()
r0[:, :, 2] .= v0

prob = ODEProblem(basic_version!, r0, (0.0, 0.1), p)

ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.08216204655659 11.05611204771623 ... 11.033013344187163 11.0143480610
2024; 11.040964658143634 11.098202251254763 ... 11.004879215341978 11.0523934
41600943; ... ; 11.059251812753288 11.076278047148032 ... 11.023336447011438 11
.036539130672317; 11.030081780523698 11.021251236452668 ... 11.07120344439117
11.076760376841124]

[12.100000000000001 12.100000000000001 ... 12.100000000000001 12.1000000000000
001; 12.100000000000001 12.100000000000001 ... 12.100000000000001 12.10000000
0000001; ... ; 12.100000000000001 12.100000000000001 ... 12.100000000000001 12.
100000000000001; 12.100000000000001 12.100000000000001 ... 12.100000000000001
12.100000000000001]

```

In this version we have encoded our initial condition to be a 3-dimensional array, with `u[:, :, 1]` being the A part and `u[:, :, 2]` being the B part.

```
@benchmark solve(prob, Tsit5())
```

```

BenchmarkTools.Trial:
  memory estimate: 186.88 MiB
  allocs estimate: 8552
  -----
  minimum time:      48.405 ms (4.47% GC)
  median time:       50.782 ms (8.46% GC)
  mean time:         50.240 ms (7.53% GC)
  maximum time:      51.030 ms (8.67% GC)
  -----
  samples:           100
  evals/sample:      1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it!

The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```

A = rand(4)
@show A

A = [0.9756009963729109, 0.13986675260334258, 0.7969695033957642, 0.9129463
070578414]

B = @view A[1:3]
B[2] = 2
@show A

A = [0.9756009963729109, 2.0, 0.7969695033957642, 0.9129463070578414]
4-element Array{Float64,1}:
 0.9756009963729109
 2.0
 0.7969695033957642
 0.9129463070578414

```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```
function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 119.55 MiB
  allocs estimate: 7082
  -----
  minimum time:      38.510 ms (5.42% GC)
  median time:      38.731 ms (5.66% GC)
  mean time:        39.168 ms (6.67% GC)
  maximum time:     41.071 ms (10.55% GC)
  -----
  samples:          128
  evals/sample:     1
```

Now, most of the allocations are taking place in `Du = D1*(Ay*u + u*Ax)` since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```
Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
```

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 29.75 MiB
  allocs estimate: 5318
  -----
  minimum time:      33.114 ms (0.00% GC)
  median time:       33.252 ms (0.00% GC)
  mean time:         33.889 ms (1.86% GC)
  maximum time:      35.613 ms (6.03% GC)
  -----
  samples:           148
  evals/sample:      1
```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability*****.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
  a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
  u = @view r[:,1]
  v = @view r[:,2]
  du = @view dr[:,1]
  dv = @view dr[:,2]
  mul!(Ayu,Ay,u)
  mul!(uAx,u,Ax)
  mul!(Ayv,Ay,v)
  mul!(vAx,v,Ax)
  @. Du = D1*(Ayu + uAx)
  @. Dv = D2*(Ayv + vAx)
  @. du = Du + a*u*u./v + ubar - α*u
  @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 29.66 MiB
  allocs estimate: 1054
  -----
  minimum time:      26.598 ms (0.00% GC)
  median time:       26.771 ms (0.00% GC)
  mean time:         27.387 ms (2.27% GC)
  maximum time:      29.134 ms (7.40% GC)
  -----
  samples:           183
  evals/sample:      1
```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
  a,α,ubar,β,D1,D2,N = p

  @inbounds for j in 2:N-1, i in 2:N-1
    du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
```

```

        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
end

@inbounds for j in 2:N-1, i in 2:N-1
    du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end

@inbounds for j in 2:N-1
    i = 1
    du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
end
@inbounds for j in 2:N-1
    i = 1
    du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end

@inbounds for i in 2:N-1
    j = 1
    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +

```

```

        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]

i = N; j = 1
du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]

i = N; j = N
du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate: 29.62 MiB
  allocs estimate: 467
  -----
  minimum time:      8.072 ms (0.00% GC)
  median time:       8.174 ms (0.00% GC)
  mean time:         8.785 ms (6.90% GC)
  maximum time:      10.455 ms (19.88% GC)
  -----
  samples:           569
  evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then **optimize our algorithm choice**. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```

prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate: 2.76 GiB
  allocs estimate: 41633
  -----
  minimum time:      1.281 s (32.53% GC)

```

```

median time:      1.460 s (43.84% GC)
mean time:        1.456 s (43.55% GC)
maximum time:     1.625 s (51.74% GC)
-----
samples:          4
evals/sample:     1

using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))

BenchmarkTools.Trial:
  memory estimate: 121.69 MiB
  allocs estimate: 20938
  -----
  minimum time:      828.984 ms (0.30% GC)
  median time:       830.989 ms (0.29% GC)
  mean time:         830.742 ms (0.29% GC)
  maximum time:      831.715 ms (0.26% GC)
  -----
  samples:           7
  evals/sample:      1

prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)

BenchmarkTools.Trial:
  memory estimate:  2.90 MiB
  allocs estimate:  77
  -----
  minimum time:     5.713 s (0.00% GC)
  median time:      5.713 s (0.00% GC)
  mean time:        5.713 s (0.00% GC)
  maximum time:     5.713 s (0.00% GC)
  -----
  samples:          1
  evals/sample:     1

@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))

BenchmarkTools.Trial:
  memory estimate:  346.41 MiB
  allocs estimate: 62155
  -----
  minimum time:     2.558 s (1.19% GC)
  median time:      2.612 s (3.21% GC)
  mean time:        2.612 s (3.21% GC)
  maximum time:     2.665 s (5.15% GC)
  -----
  samples:          2
  evals/sample:     1

```

Now let's check the allocation growth.

```

@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)

BenchmarkTools.Trial:
  memory estimate:  3.87 MiB
  allocs estimate: 51671
  -----

```



```

minimum time:      2.493 s (0.00% GC)
median time:       2.494 s (0.00% GC)
mean time:         2.494 s (0.00% GC)
maximum time:      2.494 s (0.00% GC)
-----
samples:           3
evals/sample:      1

prob = ODEProblem(fast_gm!, r0, (0.0, 500.0), p)
@benchmark solve(prob, CVODE_BDF(linear_solver=:GMRES), save_everystep=false)

BenchmarkTools.Trial:
  memory estimate:  5.28 MiB
  allocs estimate:  76217
  -----
  minimum time:     3.690 s (0.00% GC)
  median time:      3.691 s (0.00% GC)
  mean time:        3.691 s (0.00% GC)
  maximum time:     3.692 s (0.00% GC)
  -----
  samples:           2
  evals/sample:      1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is CVODE_BDF doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas CVODE_BDF is taking large steps. Additionally, the GMRES linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

This tutorial is part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq/DiffEqTutorials>

To locally run this tutorial, do the following commands:

```

using DiffEqTutorials
DiffEqTutorials.weave_file("introduction", "03-optimizing_diffeq_code.jmd")

```

Computer Information:

```
Julia Version 1.4.2
Commit 44fa15b150* (2020-05-23 18:35 UTC)
Platform Info:
  OS: Linux (x86_64-pc-linux-gnu)
  CPU: Intel(R) Core(TM) i7-9700K CPU @ 3.60GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-8.0.1 (ORCJIT, skylake)
Environment:
  JULIA_DEPOT_PATH = /builds/JuliaGPU/DiffEqTutorials.jl/.julia
  JULIA_CUDA_MEMORY_LIMIT = 2147483648
  JULIA_PROJECT = @.
  JULIA_NUM_THREADS = 4
```

Package Information:

```
Status `~/builds/JuliaGPU/DiffEqTutorials.jl/tutorials/introduction/Project.toml`
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.5.0
[0c46a032-eb83-5123-abaf-570d42b7fbaf] DifferentialEquations 6.15.0
[65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 5.3.0
[91a5bcdd-55d7-5caf-9e0b-520d859cae80] Plots 1.5.2
[90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.12.3
[c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 4.2.5
[37e2e46d-f89d-539d-b4ee-838fcccc9c8e] LinearAlgebra
```