

# Numbers with Uncertainties

Mosè Giordano, Chris Rackauckas

March 1, 2019

The result of a measurement should be given as a number with an attached uncertainties, besides the physical unit, and all operations performed involving the result of the measurement should propagate the uncertainty, taking care of correlation between quantities.

There is a Julia package for dealing with numbers with uncertainties: [Measurements.jl](#). Thanks to Julia's features, `DifferentialEquations.jl` easily works together with `Measurements.jl` out-of-the-box.

This notebook will cover some of the examples from the tutorial about classical Physics.

## 0.1 Caveat about Measurement type

Before going on with the tutorial, we must point up a subtlety of `Measurements.jl` that you should be aware of:

```
using Measurements
```

```
5.23 ± 0.14 === 5.23 ± 0.14
```

```
false
```

```
(5.23± 0.14) - (5.23 ± 0.14)
```

```
0.0 ± 0.2
```

```
(5.23 ± 0.14) / (5.23 ± 0.14)
```

```
1.0 ± 0.038
```

The two numbers above, even though have the same nominal value and the same uncertainties, are actually two different measurements that only by chance share the same figures and their difference and their ratio have a non-zero uncertainty. It is common in physics to get very similar, or even equal, results for a repeated measurement, but the two measurements are not the same thing.

Instead, if you have *one measurement* and want to perform some operations involving it, you have to assign it to a variable:

```
x = 5.23 ± 0.14
x == x
```

```
true
```

```
x - x
```

```
0.0 ± 0.0
```

```
x / x
```

```
1.0 ± 0.0
```

## 0.2 Radioactive Decay of Carbon-14

The rate of decay of carbon-14 is governed by a first order linear ordinary differential equation

$$\frac{du(t)}{dt} = -\frac{u(t)}{\tau}$$

where  $\tau$  is the mean lifetime of carbon-14, which is related to the half-life  $t_{1/2} = (5730 \pm 40)$  years by the relation  $\tau = t_{1/2} / \ln(2)$ .

```
using DifferentialEquations, Measurements, Plots
```

```
# Half-life and mean lifetime of radiocarbon, in years
```

```
t_12 = 5730 ± 40
```

```
τ = t_12 / log(2)
```

```
#Setup
```

```
u_0 = 1 ± 0
```

```
tspan = (0.0, 10000.0)
```

```
#Define the problem
```

```
radioactivedecay(u,p,t) = - u / τ
```

```
#Pass to solver
```

```
prob = ODEProblem(radioactivedecay, u_0, tspan)
```


```
sol = solve(prob, Tsit5(), reltol = 1e-8)
```

```
# Analytic solution
```

```
u = exp.(- sol.t / τ)
```

```
plot(sol.t, sol.u, label = "Numerical", xlabel = "Years", ylabel = "Fraction of  
Carbon-14")
```

```
plot!(sol.t, u, label = "Analytic")
```



The two curves are perfectly superimposed, indicating that the numerical solution matches the analytic one. We can check that also the uncertainties are correctly propagated in the numerical solution:

```
println("Quantity of carbon-14 after ", sol.t[11], " years:")
```

Quantity of carbon-14 after 5207.5228514026385 years:

```
println("Numerical: ", sol[11])
```

Numerical: 0.5326215661145899 ± 0.0023422116367677525

```
println("Analytic: ", u[11])
```

Analytic: 0.5326215654338256 ± 0.002342211664674973

Both the value of the numerical solution and its uncertainty match the analytic solution within the requested tolerance. We can also note that close to 5730 years after the beginning of the decay (half-life of the radioisotope), the fraction of carbon-14 that survived is about 0.5.

## 0.3 Simple pendulum

### 0.3.1 Small angles approximation

The next problem we are going to study is the simple pendulum in the approximation of small angles. We address this simplified case because there exists an easy analytic solution to compare.

The differential equation we want to solve is

$$\ddot{\theta} + \frac{g}{L}\theta = 0$$

where  $g = (9.79 \pm 0.02) \text{ m/s}^2$  is the gravitational acceleration measured where the experiment is carried out, and  $L = (1.00 \pm 0.01) \text{ m}$  is the length of the pendulum.

When you set up the problem for `DifferentialEquations.jl` remember to define the measurements as variables, as seen above.

```
using DifferentialEquations, Measurements, Plots

g = 9.79 ± 0.02; # Gravitational constants
L = 1.00 ± 0.01; # Length of the pendulum


#Initial Conditions
u_0 = [0 ± 0, π / 60 ± 0.01] # Initial speed and initial angle
tspan = (0.0, 6.3)

#Define the problem
function simplependulum(du,u,p,t)
    θ = u[1]
    dθ = u[2]
    du[1] = dθ
    du[2] = -(g/L)*θ
end

#Pass to solvers
prob = ODEProblem(simplependulum, u_0, tspan)
sol = solve(prob, Tsit5(), reltol = 1e-6)

# Analytic solution
u = u_0[2] .* cos.(sqrt(g / L) .* sol.t)


plot(sol.t, getindex.(sol.u, 2), label = "Numerical")
plot!(sol.t, u, label = "Analytic")
```



Also in this case there is a perfect superimposition between the two curves, including their uncertainties.

We can also have a look at the difference between the two solutions:

```
plot(sol.t, getindex(sol.u, 2) .- u, label = "")
```



## 0.4 Arbitrary amplitude

Now that we know how to solve differential equations involving numbers with uncertainties we can solve the simple pendulum problem without any approximation. This time the differential equation to solve is the following:

$$\ddot{\theta} + \frac{g}{L} \sin(\theta) = 0$$


```
g = 9.79 ± 0.02; # Gravitational constants
L = 1.00 ± 0.01; # Length of the pendulum

#Initial Conditions
u_0 = [0 ± 0, π / 3 ± 0.02] # Initial speed and initial angle
tspan = (0.0, 6.3)

#Define the problem
function simplependulum(du,u,p,t)
    θ = u[1]
    dθ = u[2]
    du[1] = dθ
    du[2] = -(g/L) * sin(θ)
end

#Pass to solvers
prob = ODEProblem(simplependulum, u_0, tspan)
sol = solve(prob, Tsit5(), reltol = 1e-6)

plot(sol.t, getindex.(sol.u, 2), label = "Numerical")
```



We note that in this case the period of the oscillations is not constant.

```
using DiffEqTutorials
DiffEqTutorials.tutorial_footer(WEAVE_ARGS[:folder],WEAVE_ARGS[:file])
```

## 0.5 Appendix

These benchmarks are part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq>

To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("type_handling","uncertainties.jmd")
```

Computer Information:

```
Julia Version 1.1.0
Commit 80516ca202 (2019-01-21 21:24 UTC)
Platform Info:
  OS: Windows (x86_64-w64-mingw32)
  CPU: Intel(R) Core(TM) i7-8700 CPU @ 3.20GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-6.0.1 (ORCJIT, skylake)
Environment:
```

```
JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.34.0\atom.exe" -a  
JULIA_NUM_THREADS = 6
```

Package Information:

```
Status `C:\Users\accou\.julia\external\DiffEqTutorials.jl\src\Project.toml`
```