

Optimizing DiffEq Code

Chris Rackauckas

March 5, 2019

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

lorenz (generic function with 1 method)

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.94 MiB
  allocs estimate: 102469
  -----
  minimum time:      3.211 ms (0.00% GC)
  median time:       6.098 ms (0.00% GC)
  mean time:         6.795 ms (42.87% GC)
  maximum time:     19.500 ms (55.84% GC)
  -----
  samples:           735
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.57 MiB
  allocs estimate: 89577
  -----
  minimum time:      2.846 ms (0.00% GC)
  median time:       3.640 ms (0.00% GC)
  mean time:         6.018 ms (42.24% GC)
  maximum time:     23.322 ms (62.07% GC)
  -----
  samples:           830
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to `***make our code non-allocating***`:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
```

```

du[2] = u[1]*(28.0-u[3]) - u[2]
du[3] = u[1]*u[2] - (8/3)*u[3]
end

```

```
lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes the internal allocations as well. When we benchmark this function, we will see quite a difference.

```

u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate:  1.37 MiB
  allocs estimate:  12964
  -----
  minimum time:     755.701 μs (0.00% GC)
  median time:      778.001 μs (0.00% GC)
  mean time:        1.203 ms (30.44% GC)
  maximum time:     11.993 ms (89.69% GC)
  -----
  samples:          4145
  evals/sample:     1

```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  6.83 KiB
  allocs estimate:  92
  -----
  minimum time:     407.100 μs (0.00% GC)
  median time:      413.800 μs (0.00% GC)
  mean time:        425.133 μs (0.55% GC)
  maximum time:     9.732 ms (95.04% GC)
  -----
  samples:          10000
  evals/sample:     1

```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```

tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  6.83 KiB
  allocs estimate:  92
  -----
  minimum time:     2.037 ms (0.00% GC)
  median time:      2.099 ms (0.00% GC)
  mean time:        2.149 ms (0.00% GC)
  maximum time:     3.032 ms (0.00% GC)
  -----
  samples:          2321
  evals/sample:     1

```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are a lot of sources online](#). But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. **structs** for example are stack-allocated "value-type"s. **Tuples** are a stack-allocated collection. The most useful data structure for DiffEq though is the **StaticArray** from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```

using StaticArrays
A = @SVector [2.0,3.0,5.0]

```

```

3-element StaticArrays.SArray{Tuple{3},Float64,1,3}:
 2.0
 3.0
 5.0

```

Notice that the 3 after **SVector** gives the size of the **SVector**. It cannot be changed. Additionally, **SVectors** are immutable, so we have to create a new **SVector** to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. **SArrays** have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons.

Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's *****optimize `lorenz` using static arrays*****. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end
```

```
lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 471.98 KiB
  allocs estimate: 2662
  -----
  minimum time:      444.599 μs (0.00% GC)
  median time:       459.900 μs (0.00% GC)
  mean time:         539.616 μs (12.50% GC)
  maximum time:      4.844 ms (90.10% GC)
  -----
  samples:           9236
  evals/sample:      1
```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 6.13 KiB
  allocs estimate: 73
  -----
  minimum time:      347.600 μs (0.00% GC)
  median time:       358.099 μs (0.00% GC)
  mean time:         365.745 μs (0.35% GC)
  maximum time:      7.172 ms (94.39% GC)
  -----
  samples:           10000
  evals/sample:      1
```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```
A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
  minimum time:     3.353 ms (0.00% GC)
  median time:      3.756 ms (0.00% GC)
  mean time:        5.473 ms (32.92% GC)
  maximum time:     15.598 ms (77.29% GC)
  -----
  samples:          912
  evals/sample:     1
```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+` `C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```
test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  5
  -----
  minimum time:     3.957 ms (0.00% GC)
```

```

median time:      4.305 ms (0.00% GC)
mean time:        5.921 ms (28.66% GC)
maximum time:     14.182 ms (60.91% GC)
-----
samples:          846
evals/sample:     1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.294 ms (0.00% GC)
median time:      3.733 ms (0.00% GC)
mean time:        5.431 ms (32.55% GC)
maximum time:     21.084 ms (83.88% GC)
-----
samples:          921
evals/sample:     1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.289 ms (0.00% GC)
median time:      3.735 ms (0.00% GC)
mean time:        5.333 ms (32.42% GC)
maximum time:     13.483 ms (74.88% GC)
-----
samples:          936
evals/sample:     1

```

is a version with only 1 array created (the output). Note that `.`s can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```
1000×1000 Array{Float64,2}:
```

```
 1.08581  1.63243  0.768384  1.11012  ...  0.493044  0.777289  0.759681
 0.496165  1.60445  1.34047  0.64133  ...  0.785945  1.12138  0.625422
 0.532528  1.43737  0.744649  1.14064  ...  1.07387  0.928458  0.397964
 0.708116  1.00732  1.3855  0.383177  ...  0.786335  1.05239  0.450632
 0.438923  0.676546  1.04783  1.61618  ...  1.30027  0.92296  1.22918
 0.410388  0.617402  0.833076  1.34194  ...  0.888788  1.38156  1.20824
 0.465104  0.791508  0.972834  1.50994  ...  1.31959  0.723721  0.936908
 0.792422  1.54326  0.737257  0.779706  ...  0.857711  0.748587  0.969454
 1.1345  0.947726  0.846407  1.58109  ...  1.37137  1.54302  0.873323
 0.843418  0.803769  0.83621  1.13932  ...  1.48331  1.2025  0.550482
 ⋮
 0.874478  1.14032  1.48811  0.645365  ...  0.971953  0.404449  1.08981
 1.15813  0.863214  0.703243  1.00293  ...  1.19418  0.767359  1.38724
 1.34819  1.28885  1.24183  1.39305  ...  0.663234  0.950089  0.998995
 1.55783  1.36547  1.00807  1.13392  ...  1.1922  1.43149  0.965352
 0.581816  1.40992  1.01032  1.35366  ...  0.614881  0.268456  0.778271
 1.52096  0.608899  1.46801  1.13741  ...  1.17421  1.39893  0.909487
 1.08899  1.12018  0.830543  0.862908  ...  0.329315  0.340049  0.628808
 1.16753  0.398144  1.16579  0.714621  ...  0.350726  0.585933  0.520727
 0.71317  1.245  1.04066  0.60053  ...  0.835138  0.51503  1.66689
```

Also, the @. macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```
BenchmarkTools.Trial:
```

```
memory estimate: 7.63 MiB
```

```
allocs estimate: 3
```

```
-----
```

```
minimum time: 3.281 ms (0.00% GC)
```

```
median time: 3.594 ms (0.00% GC)
```

```
mean time: 5.199 ms (31.96% GC)
```

```
maximum time: 14.682 ms (75.17% GC)
```

```
-----
```

```
samples: 960
```

```
evals/sample: 1
```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via .=. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:


```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.591 ms (0.00% GC)
  median time:       1.990 ms (0.00% GC)
  mean time:         2.081 ms (0.00% GC)
  maximum time:      3.896 ms (0.00% GC)
  -----
  samples:           2385
  evals/sample:      1
```

If we use `@.` before the `=`, then it will turn it into `.=`:

```
test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.589 ms (0.00% GC)
  median time:       1.823 ms (0.00% GC)
  mean time:         1.910 ms (0.00% GC)
  maximum time:      7.595 ms (0.00% GC)
  -----
  samples:           2608
  evals/sample:      1
```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```
test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 32 bytes
  allocs estimate: 1
  -----
  minimum time:      1.731 ms (0.00% GC)
  median time:       1.926 ms (0.00% GC)
  mean time:         1.998 ms (0.00% GC)
  maximum time:      4.208 ms (0.00% GC)
  -----
  samples:           2489
  evals/sample:      1
```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:      12.658 ms (0.00% GC)
  median time:       14.366 ms (0.00% GC)
  mean time:         15.908 ms (11.32% GC)
  maximum time:      22.053 ms (25.14% GC)
  -----
  samples:           314
  evals/sample:      1
```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```
using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:      11.806 ms (0.00% GC)
  median time:       14.408 ms (0.00% GC)
  mean time:         15.375 ms (0.00% GC)
  maximum time:      29.375 ms (0.00% GC)
  -----
  samples:           325
  evals/sample:      1
```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, ***we can fuse together higher level linear algebra operations using BLAS***. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal{Float64,3}}{1.0 for i in 1:N-1, [-2.0 for i in 1:N], [1.0 for i in 1:N-1]})
Ay = copy(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du ./ a.*u.*u ./ v ./ ubar ./ α*u
    dr[:, :, 2] = Dv ./ a.*u.*u ./ β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64}(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand{Float64}()
r0[:, :, 2] .= vss

prob = ODEProblem{basic_version!,r0,(0.0,0.1),p}
```

```
ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.0009 11.0233 ... 11.0931 11.0092; 11.0634 11.0567 ... 11.0242 11.0217;
... ; 11.0562 11.0953 ... 11.0292 11.0947; 11.0742 11.0155 ... 11.0931 11.0769]

[12.1 12.1 ... 12.1 12.1; 12.1 12.1 ... 12.1 12.1; ... ; 12.1 12.1 ... 12.1 12.1; 1
2.1 12.1 ... 12.1 12.1]
```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 186.88 MiB
  allocs estimate: 8589
  -----
  minimum time:      108.349 ms (23.84% GC)
  median time:       234.283 ms (63.46% GC)
  mean time:         206.111 ms (58.55% GC)
```

```

maximum time:      296.936 ms (69.61% GC)
-----
samples:          25
evals/sample:     1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```

A = rand(4)
@show A

```

```

A = [0.945066, 0.654731, 0.238168, 0.75385]

```

```

B = @view A[1:3]
B[2] = 2
@show A

```

```

A = [0.945066, 2.0, 0.238168, 0.75385]
4-element Array{Float64,1}:
 0.9450656074747363
 2.0
 0.2381684081695823
 0.7538502799057683

```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```

function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(A*y*u + u*Ax)
    Dv = D2*(A*y*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 119.55 MiB
  allocs estimate: 7119
  -----
  minimum time:      78.664 ms (20.77% GC)
  median time:       141.704 ms (52.31% GC)
  mean time:         168.839 ms (60.34% GC)
  maximum time:      277.781 ms (75.07% GC)
  -----
  samples:           30
  evals/sample:      1

```

Now, most of the allocations are taking place in $Du = D1*(Ay*u + u*Ax)$ since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```

Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
  a,α,ubar,β,D1,D2 = p
  u = @view r[:, :, 1]
  v = @view r[:, :, 2]
  du = @view dr[:, :, 1]
  dv = @view dr[:, :, 2]
  mul!(Ayu,Ay,u)
  mul!(uAx,u,Ax)
  mul!(Ayv,Ay,v)
  mul!(vAx,v,Ax)
  @. Du = D1*(Ayu + uAx)
  @. Dv = D2*(Ayv + vAx)
  @. du = Du + a*u*u./v + ubar - α*u
  @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 29.76 MiB
  allocs estimate: 5355
  -----
  minimum time:      51.246 ms (6.35% GC)
  median time:       53.395 ms (5.81% GC)
  mean time:         54.372 ms (7.63% GC)
  maximum time:      71.935 ms (21.71% GC)
  -----
  samples:           92
  evals/sample:      1

```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's

easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability***.**

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
    a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 29.66 MiB
  allocs estimate: 1090
  -----
  minimum time:      43.197 ms (7.18% GC)
  median time:       45.580 ms (6.47% GC)
  mean time:         46.728 ms (8.30% GC)
  maximum time:      53.209 ms (8.10% GC)
  -----
  samples:           108
  evals/sample:      1
```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a,α,ubar,β,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
                    a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
```

```

end
@inbounds for j in 2:N-1
    i = 1
    du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds for i in 2:N-1
    j = 1
    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +

```

```

        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 29.63 MiB
  allocs estimate: 505
  -----
  minimum time:      10.661 ms (21.82% GC)
  median time:       11.512 ms (24.29% GC)
  mean time:         12.154 ms (28.71% GC)
  maximum time:      23.696 ms (41.68% GC)
  -----
  samples:           412
  evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then *****optimize our algorithm choice*****. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```

prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 2.76 GiB
  allocs estimate: 41671
  -----
  minimum time:      2.243 s (33.88% GC)
  median time:       12.423 s (71.61% GC)
  mean time:         12.423 s (71.61% GC)
  maximum time:      22.602 s (75.35% GC)
  -----
  samples:           2
  evals/sample:      1

```



```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  48.71 MiB
  allocs estimate:  12296
  -----
  minimum time:     206.827 ms (2.95% GC)
  median time:      211.853 ms (3.02% GC)
  mean time:        225.522 ms (9.38% GC)
  maximum time:     335.304 ms (38.61% GC)
  -----
  samples:          23
  evals/sample:     1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  2.91 MiB
  allocs estimate:  113
  -----
  minimum time:     4.733 s (0.00% GC)
  median time:      4.804 s (0.00% GC)
  mean time:        4.804 s (0.00% GC)
  maximum time:     4.875 s (0.00% GC)
  -----
  samples:          2
  evals/sample:     1
```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  257.14 MiB
  allocs estimate:  72389
  -----
  minimum time:     1.389 s (9.72% GC)
  median time:      1.435 s (14.02% GC)
  mean time:        1.498 s (17.56% GC)
  maximum time:     1.733 s (29.69% GC)
  -----
  samples:          4
  evals/sample:     1
```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  3.44 MiB
  allocs estimate:  62253
  -----
  minimum time:     1.190 s (0.00% GC)
  median time:      1.197 s (0.00% GC)
  mean time:        1.196 s (0.00% GC)
  maximum time:     1.205 s (0.00% GC)
  -----
  samples:          5
  evals/sample:     1

prob = ODEProblem(fast_gm!,r0,(0.0,500.0),p)
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  4.27 MiB
  allocs estimate:  82936
  -----
  minimum time:     1.580 s (0.00% GC)
  median time:      1.585 s (0.00% GC)
  mean time:        1.585 s (0.00% GC)
  maximum time:     1.589 s (0.00% GC)
  -----
  samples:          4
  evals/sample:     1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is CVODE_BDF doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas CVODE_BDF is taking large steps. Additionally, the GMRES linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

```

using DiffEqTutorials
DiffEqTutorials.tutorial_footer(WEAVE_ARGS[:folder],WEAVE_ARGS[:file])

```

0.4 Appendix

These benchmarks are part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq>

To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("introduction","optimizing_diffeq_code.jmd")
```

Computer Information:

```
Julia Version 1.1.0
Commit 80516ca202 (2019-01-21 21:24 UTC)
Platform Info:
  OS: Windows (x86_64-w64-mingw32)
  CPU: Intel(R) Core(TM) i7-8700 CPU @ 3.20GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-6.0.1 (ORCJIT, skylake)
Environment:
  JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.34.0\atom.exe" -a
  JULIA_NUM_THREADS = 6
```

Package Information:

```
Status `C:\Users\accou\.julia\external\DiffEqTutorials.jl\Project.toml`
[7e558dbc-694d-5a72-987c-6f4ebed21442] ArbNumerics 0.3.6
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.4.2
[be33ccc6-a3ff-5ff2-a52e-74243cff1e17] CUDAnative 1.0.1
[3a865a2d-5b23-5a0f-bc46-62713ec82fae] CuArrays 0.9.1
[55939f99-70c6-5e9b-8bb0-5071ed7d61fd] DecFP 0.4.8
[abce61dc-4473-55a0-ba07-351d65e31d42] Decimals 0.4.0
[459566f4-90b8-5000-8ac3-15dfb0a30def] DiffEqCallbacks 2.5.2
[f3b72e0c-5b89-59e1-b016-84e28bfd966d] DiffEqDevTools 2.6.1
[1130ab10-4a5a-5621-a13d-e4788d82bd4c] DiffEqParamEstim 1.6.0
[055956cb-9e8b-5191-98cc-73ae4a59e68a] DiffEqPhysics 3.1.0
[0c46a032-eb83-5123-abaf-570d42b7fbaf] DifferentialEquations 6.3.0
[497a8b3b-efae-58df-a0af-a86822472b78] DoubleFloats 0.7.5
[f6369f11-7733-5829-9624-2563aa707210] ForwardDiff 0.10.3
[7073ff75-c697-5162-941a-fcdaad2a7d2a] IJulia 1.17.0
[eff96d63-e80a-5855-80a2-b1b0885c5ab7] Measurements 2.0.0
[429524aa-4258-5aef-a3af-852621145aeb] Optim 0.17.2
[1dea7af3-3e70-54e6-95c3-0bf5283fa5ed] OrdinaryDiffEq 5.3.0
[65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 4.1.1
[91a5bcdd-55d7-5caf-9e0b-520d859cae80] Plots 0.23.1
[731186ca-8d62-57ce-b412-fbd966d074cd] RecursiveArrayTools 0.20.0
[90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.10.3
[c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 3.1.0
```

[1986cc42-f94f-5a68-af5c-568840ba703d] Unitful 0.14.0
[2a06ce6d-1589-592b-9c33-f37faeaed826] UnitfulPlots 0.0.0
[44d3d7a6-8a23-5bf8-98c5-b353f8df5ec9] Weave 0.7.2
[b77e0a4c-d291-57a0-90e8-8db25a27a240] InteractiveUtils
[37e2e46d-f89d-539d-b4ee-838fcccc9c8e] LinearAlgebra
[44cfe95a-1eb2-52ea-b672-e2afdf69b78f] Pkg