

Optimizing DiffEq Code

Chris Rackauckas

March 1, 2019

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.94 MiB
  allocs estimate: 102470
  -----
  minimum time:      3.351 ms (0.00% GC)
  median time:       5.176 ms (0.00% GC)
  mean time:         7.153 ms (44.36% GC)
  maximum time:     13.009 ms (56.09% GC)
  -----
  samples:           699
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.57 MiB
  allocs estimate: 89578
  -----
  minimum time:      2.962 ms (0.00% GC)
  median time:       3.562 ms (0.00% GC)
  mean time:         6.257 ms (44.17% GC)
  maximum time:     12.294 ms (62.63% GC)
  -----
  samples:           799
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to `***make our code non-allocating***`:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
```

```

du[2] = u[1]*(28.0-u[3]) - u[2]
du[3] = u[1]*u[2] - (8/3)*u[3]
end

```

```
lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes the internal allocations as well. When we benchmark this function, we will see quite a difference.

```

u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate:  1.37 MiB
  allocs estimate:  12965
  -----
  minimum time:     758.900 μs (0.00% GC)
  median time:      778.500 μs (0.00% GC)
  mean time:        1.207 ms (30.69% GC)
  maximum time:     8.189 ms (88.54% GC)
  -----
  samples:          4137
  evals/sample:     1

```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  6.91 KiB
  allocs estimate:  93
  -----
  minimum time:     408.900 μs (0.00% GC)
  median time:      413.600 μs (0.00% GC)
  mean time:        424.030 μs (0.53% GC)
  maximum time:     8.055 ms (94.31% GC)
  -----
  samples:          10000
  evals/sample:     1

```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```

tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  6.91 KiB
  allocs estimate:  93
  -----
  minimum time:     2.086 ms (0.00% GC)
  median time:      2.109 ms (0.00% GC)
  mean time:        2.142 ms (0.00% GC)
  maximum time:     2.646 ms (0.00% GC)
  -----
  samples:          2333
  evals/sample:     1

```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are a lot of sources online](#). But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. **structs** for example are stack-allocated "value-type"s. **Tuples** are a stack-allocated collection. The most useful data structure for DiffEq though is the **StaticArray** from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```

using StaticArrays
A = @SVector [2.0,3.0,5.0]

```

```

3-element StaticArrays.SArray{Tuple{3},Float64,1,3}:
 2.0
 3.0
 5.0

```

Notice that the 3 after **SVector** gives the size of the **SVector**. It cannot be changed. Additionally, **SVectors** are immutable, so we have to create a new **SVector** to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. **SArrays** have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons.

Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's *****optimize `lorenz` using static arrays*****. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end
```

```
lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 472.06 KiB
  allocs estimate: 2663
  -----
  minimum time:      446.700 μs (0.00% GC)
  median time:       452.500 μs (0.00% GC)
  mean time:         543.311 μs (14.26% GC)
  maximum time:      5.149 ms (87.38% GC)
  -----
  samples:           9171
  evals/sample:      1
```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 6.20 KiB
  allocs estimate: 74
  -----
  minimum time:      354.200 μs (0.00% GC)
  median time:       359.200 μs (0.00% GC)
  mean time:         368.543 μs (0.39% GC)
  maximum time:      7.641 ms (94.81% GC)
  -----
  samples:           10000
  evals/sample:      1
```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```
A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
  minimum time:     3.410 ms (0.00% GC)
  median time:      3.768 ms (0.00% GC)
  mean time:        5.478 ms (33.60% GC)
  maximum time:     10.928 ms (56.54% GC)
  -----
  samples:          912
  evals/sample:     1
```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+` `C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```
test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  5
  -----
  minimum time:     4.098 ms (0.00% GC)
```

```

median time:      4.500 ms (0.00% GC)
mean time:        6.190 ms (29.59% GC)
maximum time:     12.665 ms (59.66% GC)
-----
samples:          807
evals/sample:     1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.371 ms (0.00% GC)
median time:      3.705 ms (0.00% GC)
mean time:        5.404 ms (33.71% GC)
maximum time:     12.114 ms (65.73% GC)
-----
samples:          924
evals/sample:     1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.373 ms (0.00% GC)
median time:      3.734 ms (0.00% GC)
mean time:        5.430 ms (33.62% GC)
maximum time:     10.643 ms (64.01% GC)
-----
samples:          919
evals/sample:     1

```

is a version with only 1 array created (the output). Note that `.`s can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```
1000×1000 Array{Float64,2}:
 0.782396  1.27273  0.99945  0.798116  ...  1.60205  1.29474  1.40659
 1.27593   1.30831  0.813798 0.786518  ...  1.33909  0.717149 0.255474
 0.669986  1.05846  0.387261 0.332641  ...  0.832098 0.835468 1.14676
 0.8415    0.796969 1.27448  1.23946   ...  1.06287 0.579353 0.594012
 1.57699   0.895062 0.495596 0.762927  ...  0.782056 0.834771 1.25651
 0.26711   1.38793  1.30185  0.80818   ...  1.12392  0.910387 0.355808
 1.09234   1.28967  0.634475 1.43136   ...  0.99466  0.93591  1.12888
 1.23446   1.52558  0.75481  1.09524   ...  0.968433 1.05569  1.00855
 1.57969   1.23697  1.22336  0.612308  ...  0.524088 0.811354 0.153427
 0.449865  1.5533   0.210948 0.632964  ...  1.24475  0.846527 0.986542
 ⋮
 0.816875  0.981547 1.35322  0.842467  ...  0.669272 1.29665  0.506794
 1.44297   1.09976  1.60576  1.32895   ...  0.69796  0.303002 0.971966
 0.880681  0.680316 0.574889 1.00186   ...  1.26447  0.887937 1.40566
 0.735227  1.11198  1.51583  0.650795  ...  1.26516  0.38249  0.626055
 1.10927   1.58868  0.720827 0.656152  ...  0.383016 0.45485  0.730617
 1.46155   1.02671  1.34931  1.31096   ...  0.495825 0.224874 0.898988
 1.14988   0.28851  1.05487  0.801554  ...  0.733034 0.391091 1.32746
 0.894077  0.834035 0.840777 0.858741  ...  0.855585 0.762181 0.755002
 1.66896   1.33373  1.27515  0.845089  ...  0.727711 1.17815  0.0979067
```

Also, the @. macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```
BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
 minimum time:     3.397 ms (0.00% GC)
 median time:      3.732 ms (0.00% GC)
 mean time:        5.410 ms (33.42% GC)
 maximum time:     10.213 ms (64.92% GC)
  -----
 samples:          924
 evals/sample:     1
```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via .=. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:


```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.579 ms (0.00% GC)
  median time:       1.846 ms (0.00% GC)
  mean time:         1.897 ms (0.00% GC)
  maximum time:      3.653 ms (0.00% GC)
  -----
  samples:           2623
  evals/sample:      1
```

If we use `@.` before the `=`, then it will turn it into `.=`:

```
test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      1.588 ms (0.00% GC)
  median time:       1.857 ms (0.00% GC)
  mean time:         1.915 ms (0.00% GC)
  maximum time:      3.311 ms (0.00% GC)
  -----
  samples:           2599
  evals/sample:      1
```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```
test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 32 bytes
  allocs estimate: 1
  -----
  minimum time:      1.703 ms (0.00% GC)
  median time:       2.049 ms (0.00% GC)
  mean time:         2.225 ms (0.00% GC)
  maximum time:      8.359 ms (0.00% GC)
  -----
  samples:           2239
  evals/sample:      1
```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:      13.001 ms (0.00% GC)
  median time:       15.043 ms (0.00% GC)
  mean time:         16.701 ms (11.71% GC)
  maximum time:      22.711 ms (30.64% GC)
  -----
  samples:           299
  evals/sample:      1
```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```
using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:      11.912 ms (0.00% GC)
  median time:       13.765 ms (0.00% GC)
  mean time:         13.845 ms (0.00% GC)
  maximum time:      17.390 ms (0.00% GC)
  -----
  samples:           361
  evals/sample:      1
```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, ***we can fuse together higher level linear algebra operations using BLAS***. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal}([1.0 for i in 1:N-1],[-2.0 for i in 1:N],[1.0 for i in 1:N-1]))
Ay = copy(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du ./ a.*u.*u ./ v ./ ubar ./ α*u
    dr[:, :, 2] = Dv ./ a.*u.*u ./ β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64}(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand{Float64}()
r0[:, :, 2] .= vss

prob = ODEProblem{Float64}(basic_version!,r0,(0.0,0.1),p)
```

```
ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.0683 11.0412 ... 11.0972 11.0833; 11.0894 11.0801 ... 11.0804 11.07; ...
; 11.0789 11.0721 ... 11.0654 11.0146; 11.0474 11.0173 ... 11.0371 11.0037]

[12.1 12.1 ... 12.1 12.1; 12.1 12.1 ... 12.1 12.1; ... ; 12.1 12.1 ... 12.1 12.1; 1
2.1 12.1 ... 12.1 12.1]
```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 194.52 MiB
  allocs estimate: 8938
  -----
  minimum time:      107.826 ms (27.51% GC)
  median time:       296.504 ms (72.67% GC)
  mean time:         218.174 ms (62.44% GC)
```

```

maximum time:      303.086 ms (71.37% GC)
-----
samples:          23
evals/sample:     1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```

A = rand(4)
@show A

```

```

A = [0.983385, 0.856037, 0.920511, 0.562053]

```

```

B = @view A[1:3]
B[2] = 2
@show A

```

```

A = [0.983385, 2.0, 0.920511, 0.562053]
4-element Array{Float64,1}:
 0.9833850123250132
 2.0
 0.9205113078007803
 0.562053483913604

```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```

function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 124.45 MiB
  allocs estimate: 7408
  -----
  minimum time:     95.363 ms (18.31% GC)
  median time:      153.662 ms (54.56% GC)
  mean time:        180.262 ms (61.99% GC)
  maximum time:     243.469 ms (73.25% GC)
  -----
  samples:          29
  evals/sample:     1

```

Now, most of the allocations are taking place in $Du = D1*(Ay*u + u*Ax)$ since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```

Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
  a,α,ubar,β,D1,D2 = p
  u = @view r[:, :, 1]
  v = @view r[:, :, 2]
  du = @view dr[:, :, 1]
  dv = @view dr[:, :, 2]
  mul!(Ayu,Ay,u)
  mul!(uAx,u,Ax)
  mul!(Ayv,Ay,v)
  mul!(vAx,v,Ax)
  @. Du = D1*(Ayu + uAx)
  @. Dv = D2*(Ayv + vAx)
  @. du = Du + a*u*u./v + ubar - α*u
  @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 30.98 MiB
  allocs estimate: 5572
  -----
  minimum time:     53.604 ms (6.44% GC)
  median time:      57.521 ms (6.29% GC)
  mean time:        57.635 ms (8.29% GC)
  maximum time:     65.138 ms (12.03% GC)
  -----
  samples:          87
  evals/sample:     1

```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's

easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability***.**

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
    a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 30.88 MiB
  allocs estimate: 1133
  -----
  minimum time:      47.702 ms (6.95% GC)
  median time:       53.287 ms (7.41% GC)
  mean time:         54.605 ms (9.85% GC)
  maximum time:      72.550 ms (19.02% GC)
  -----
  samples:           92
  evals/sample:      1
```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a,α,ubar,β,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
                    a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
```

```

end
@inbounds for j in 2:N-1
    i = 1
    du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds for i in 2:N-1
    j = 1
    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +

```

```

        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 30.85 MiB
  allocs estimate: 524
  -----
  minimum time:      12.084 ms (25.19% GC)
  median time:       13.079 ms (26.69% GC)
  mean time:         14.043 ms (32.49% GC)
  maximum time:      24.522 ms (44.27% GC)
  -----
  samples:           356
  evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then *****optimize our algorithm choice*****. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```

prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 2.76 GiB
  allocs estimate: 41690
  -----
  minimum time:      3.299 s (26.72% GC)
  median time:       17.577 s (34.29% GC)
  mean time:         17.577 s (34.29% GC)
  maximum time:      31.855 s (35.07% GC)
  -----
  samples:           2
  evals/sample:      1

```



```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate: 118.66 MiB
  allocs estimate: 33065
  -----
  minimum time:      639.668 ms (4.12% GC)
  median time:      753.785 ms (28.07% GC)
  mean time:        750.719 ms (25.56% GC)
  maximum time:      862.338 ms (34.10% GC)
  -----
  samples:           7
  evals/sample:      1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 2.91 MiB
  allocs estimate: 114
  -----
  minimum time:      5.105 s (0.00% GC)
  median time:      5.105 s (0.00% GC)
  mean time:        5.105 s (0.00% GC)
  maximum time:      5.105 s (0.00% GC)
  -----
  samples:           1
  evals/sample:      1
```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate: 343.59 MiB
  allocs estimate: 93104
  -----
  minimum time:      1.735 s (0.00% GC)
  median time:      1.822 s (10.62% GC)
  mean time:        1.897 s (11.79% GC)
  maximum time:      2.134 s (22.37% GC)
  -----
  samples:           3
  evals/sample:      1
```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  4.26 MiB
  allocs estimate:  82710
  -----
  minimum time:     1.560 s (0.00% GC)
  median time:      1.569 s (0.00% GC)
  mean time:        1.568 s (0.00% GC)
  maximum time:     1.574 s (0.00% GC)
  -----
  samples:          4
  evals/sample:     1

prob = ODEProblem(fast_gm!, r0, (0.0, 500.0), p)
@benchmark solve(prob, CVODE_BDF(linear_solver=:GMRES), save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  5.55 MiB
  allocs estimate:  114354
  -----
  minimum time:     2.153 s (0.00% GC)
  median time:      2.193 s (0.00% GC)
  mean time:        2.193 s (0.00% GC)
  maximum time:     2.234 s (0.00% GC)
  -----
  samples:          3
  evals/sample:     1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is CVODE_BDF doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas CVODE_BDF is taking large steps. Additionally, the GMRES linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

```

using DiffEqTutorials
DiffEqTutorials.tutorial_footer(WEAVE_ARGS[:folder], WEAVE_ARGS[:file])

```

0.4 Appendix

These benchmarks are part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq>

To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("introduction","optimizing_diffeq_code.jmd")
```

Computer Information:

```
Julia Version 1.1.0
Commit 80516ca202 (2019-01-21 21:24 UTC)
Platform Info:
  OS: Windows (x86_64-w64-mingw32)
  CPU: Intel(R) Core(TM) i7-8700 CPU @ 3.20GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-6.0.1 (ORCJIT, skylake)
Environment:
  JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.34.0\atom.exe" -a
  JULIA_NUM_THREADS = 6
```

Package Information:

```
Status `C:\Users\accou\.julia\environments\v1.1\Project.toml`
[7e558dbc-694d-5a72-987c-6f4ebcd21442] ArbNumerics 0.3.6
[c52e3926-4ff0-5f6e-af25-54175e0327b1] Atom 0.7.14
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.4.2
[336ed68f-0bac-5ca0-87d4-7b16caf5d00b] CSV 0.4.3
[3895d2a7-ec45-59b8-82bb-cfc6a382f9b3] CUDAapi 0.6.0
[be33ccc6-a3ff-5ff2-a52e-74243cff1e17] CUDAnative 1.0.1
[3a865a2d-5b23-5a0f-bc46-62713ec82fae] CuArrays 0.9.1
[a93c6f00-e57d-5684-b7b6-d8193f3e46c0] DataFrames 0.17.1
[55939f99-70c6-5e9b-8bb0-5071ed7d61fd] DecFP 0.4.8
[abce61dc-4473-55a0-ba07-351d65e31d42] Decimals 0.4.0
[bcd4f6db-9728-5f36-b5f7-82caef46ccdb] DelayDiffEq 5.2.0+
[39dd38d3-220a-591b-8e3c-4c3a8c710a94] Dierckx 0.4.1
[2b5f629d-d688-5b77-993f-72d75c75574e] DiffEqBase 5.4.0+
[bb2cbb15-79fc-5d1e-9bf1-8ae49c7c1650] DiffEqBenchmarks 0.0.0
[459566f4-90b8-5000-8ac3-15dfb0a30def] DiffEqCallbacks 2.5.2
[f3b72e0c-5b89-59e1-b016-84e28bfd966d] DiffEqDevTools 2.6.1
[aae7a2af-3d4f-5e19-a356-7da93b79d9d0] DiffEqFlux 0.2.0
[c894b116-72e5-5b58-be3c-e6d8d4ac2b12] DiffEqJump 6.1.0+
[1130ab10-4a5a-5621-a13d-e4788d82bd4c] DiffEqParamEstim 1.6.0+
[055956cb-9e8b-5191-98cc-73ae4a59e68a] DiffEqPhysics 3.1.0
[a077e3f3-b75c-5d7f-a0c6-6bc4c8ec64a9] DiffEqProblemLibrary 4.1.0
[225cb15b-72e6-54e6-9a40-306d353791de] DiffEqTutorials 0.0.0
```

[0c46a032-eb83-5123-abaf-570d42b7fbaa] DifferentialEquations 6.3.0
 [497a8b3b-efae-58df-a0af-a86822472b78] DoubleFloats 0.7.5
 [587475ba-b771-5e3f-ad9e-33799f191a9c] Flux 0.7.3
 [f6369f11-7733-5829-9624-2563aa707210] ForwardDiff 0.10.3+
 [28b8d3ca-fb5f-59d9-8090-bfdbd6d07a71] GR 0.38.1
 [7073ff75-c697-5162-941a-fcdaad2a7d2a] IJulia 1.17.0
 [c601a237-2ae4-5e1e-952c-7a85b0c7eef1] Interact 0.9.1
 [b6b21f68-93f8-5de0-b562-5493be1d77c9] Ipopt 0.5.4
 [4076af6c-e467-56ae-b986-b466b2749572] JuMP 0.19.0
 [e5e0dc1b-0480-54bc-9374-aad01c23163d] Juno 0.5.4
 [7f56f5a3-f504-529b-bc02-0b1fe5e64312] LSODA 0.4.0
 [eff96d63-e80a-5855-80a2-b1b0885c5ab7] Measurements 2.0.0
 [76087f3c-5699-56af-9a33-bf431cd00edd] NLOpt 0.5.1
 [c030b06c-0b6d-57c2-b091-7029874bd033] ODE 2.4.0
 [54ca160b-1b9f-5127-a996-1867f4bc2a2c] ODEInterface 0.4.5+
 [09606e27-ecf5-54fc-bb29-004bd9f985bf] ODEInterfaceDiffEq 3.0.0
 [429524aa-4258-5aef-a3af-852621145aeb] Optim 0.17.2
 [1dea7af3-3e70-54e6-95c3-0bf5283fa5ed] OrdinaryDiffEq 5.3.0+
 [65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 4.1.1
 [91a5bcdd-55d7-5caf-9e0b-520d859cae80] Plots 0.23.0
 [71ad9d73-34c4-5ce9-b7b1-f7bd31ac38ba] PuMaS 0.0.0
 [d330b81b-6aea-500a-939a-2ce795aea3ee] PyPlot 2.7.0
 [731186ca-8d62-57ce-b412-fbd966d074cd] RecursiveArrayTools 0.20.0
 [90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.10.2
 [789caeaf-c7a9-5a7d-9973-96adeb23e2a0] StochasticDiffEq 6.1.1+
 [c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 3.1.0+
 [1986cc42-f94f-5a68-af5c-568840ba703d] Unitful 0.14.0
 [2a06ce6d-1589-592b-9c33-f37faeaed826] UnitfulPlots 0.0.0
 [44d3d7a6-8a23-5bf8-98c5-b353f8df5ec9] Weave 0.7.2